**Math 174**
**Calculus II**

**Dr. Constant J. Goutziers**
**Department of Mathematics, Computer Science and Statistics**
**goutzicj@oneonta.edu**

# Lesson 24

## Areas and Lengths in Polar Coordinates

# Initializations

```
> restart;
  with(plots):
>
```

# 24.1 Areas in Polar Coordinates

The area $A$ of the region bounded by

$$\theta = \theta_1,\ \theta = \theta_2, \text{ and } r = r(\theta)$$

is given by

$$A = \frac{1}{2}\int_{\theta_1}^{\theta_2} r^2(\theta)\, d\theta$$

Mathematical details will be provided in class.

## Examples

### *Example 24.1.1*

**Find the area of the region enclosed by the curves**

$$\theta = \frac{\pi}{6},\ \theta = \frac{2\pi}{5}, \text{ and } r = 3\,\theta$$

**Solution**
First sketch the region with pencil and paper. An accurate plot is given below.




Code and evaluate the integral that represents the desired area.

```
> f:=theta->3*theta;
  e1:=Int(1/2*f(theta)^2, theta=Pi/6..2*Pi/5);
```

$$f := \theta \rightarrow 3\,\theta$$

$$e1 := \int_{\frac{1}{6}\pi}^{\frac{2}{5}\pi} \frac{9}{2}\,\theta^2\,d\theta \qquad \textbf{(2.1.1.1)}$$

```
> area:=value(e1);
  evalf(area);
```

$$area := \frac{1603}{18000}\,\pi^3$$

$$2.761281196 \qquad \textbf{(2.1.1.2)}$$

```
>
```

*Example 24.1.2*

**Compute the area of one lobe of the rose**

$$r = \cos 4\theta$$

**Solution**

Plot the rose.

```
> f:=theta->cos(4*theta);
```

```
polarplot(f(theta), theta=0..2*Pi, color=blue, scaling=
constrained);
```

$$f := \theta \rightarrow \cos(4\,\theta)$$




Clearly, the plot has eight lobes, each of which corresponds to a theta-interval of length $\frac{2\pi}{8} = \frac{\pi}{4}$. We plot one such lobe.

```
> polarplot(f(theta), theta=-Pi/8..Pi/8, color=red);
```




The area of this lobe is given by

```
> e1:=Int(1/2*f(theta)^2, theta=-Pi/8..Pi/8);
```

$$e1 := \int_{-\frac{1}{8}\pi}^{\frac{1}{8}\pi} \frac{1}{2}\cos(4\,\theta)^2\,d\theta \qquad \textbf{(2.1.2.1)}$$

```
> area:=value(e1);
evalf(area);
```

$$area := \frac{1}{16}\pi$$

$$0.1963495409 \qquad (2.1.2.2)$$

>

## 24.2 Arc Length in Polar Coordinates

The arc length of a curve in polar coordinates

$$r = r(\theta) \qquad \theta_1 \le \theta \le \theta_2$$

is given by

$$\int_{\theta_1}^{\theta_2} \sqrt{r^2(\theta) + (r'(\theta))^2}\, d\theta$$

Mathematical details will be provided in class.

### Examples

#### *Example 24.2.1*

**Find the length of the top half of the cardioid**

$$\boldsymbol{r = 1 - \cos\theta}$$

**Solution**

First plot of the curve.

```
> f:=theta->1-cos(theta);
polarplot(f(theta), theta=0..Pi, color=red, scaling=
constrained);
```

$$f := \theta \rightarrow 1 - \cos(\theta)$$




Code and evaluate the integral which represents the arc length of this curve.

```
> e1:=Int(sqrt(f(theta)^2+D(f)(theta)^2), theta=0..Pi);
e2:=simplify(e1);
```

$$e1 := \int_0^{\pi} \sqrt{(1 - \cos(\theta))^2 + \sin(\theta)^2}\, d\theta$$

$$e2 := \int_0^{\pi} \sqrt{2 - 2\cos(\theta)}\, d\theta \tag{3.1.1.1}$$

```
> ans:=value(e2);
```

$$ans := 4 \tag{3.1.1.2}$$

As always, exact evaluation of integrals corresponding to arc lengths can be very difficult, if not impossible. If exact integration fails we can obtain a decimal approximation by using a numerical method.

```
>
```